C列にB列の累計をsumifとrow関数を使って入れています。

```

    =ARRAYFORMULA(if(B2:B="","",sumif(row(B2:B),"<="&row(B2:B),B2:B)))
    
```

現状だとA列の日付が3月でも4月関係なく累計金額が算出されるようになっております。

これを月が変わったら￥0からまたスタートするようにしたいです。

関数の知識が乏しいのですが、sumifの条件にmonth(B2:B)など入れたりしてたのですが、どうもうまくいかずご質問させて頂きました。  
何かアドバイスをいただけるとありがたいです。

[![画像の説明をここに入力](https://i.stack.imgur.com/6qwWu.png)](https://i.stack.imgur.com/6qwWu.png)
CC BY-SA 4.0
2022-04-19T04:31:38.703
88394
2022-04-29T14:26:59.777
2022-04-19T04:49:18.900
3060
51218
question
google-spreadsheet
月ごとの累計の出し方について
A列の日付が昇順に並んでいるなら以下で可能です。

```

    =ARRAYFORMULA(IF(B2:B="",,SUMIF(ROW(B2:B),"<="&ROW(B2:B),B2:B)-SUMIF(YEAR(A2:A)*100+MONTH(A2:A),"<"&YEAR(A2:A)*100+MONTH(A2:A),B2:B)))
    
```
CC BY-SA 4.0
2022-04-29T14:26:59.777
88569
2022-04-29T14:26:59.777
52409
88394
answer